
Objetivos

◆ Comprender los problemas relacionados con la 
comunicación entre procesos concurrentes 

◆ Revisar los mecanismos de sincronización que se utilizan 
para la comunicación mediante datos compartidos entre 
tareas 

◆ Estudiar la forma de realizar este tipo de comunicación 
en Ada

Comunicación y sincronización

◆ Raras veces los procesos de un sistema son 
independientes unos de otros
– A menudo cooperan para un fin común o compiten por la utilización 

de recursos.
◆ Para ello es necesario realizar operaciones de 

comunicación y sincronización entre procesos
– dos procesos se comunican cuando hay una transferencia de 

información de uno a otro
– dos procesos están sincronizados cuando hay restricciones en el 

orden en que ejecutan algunas de sus acciones

Los dos conceptos están relacionados
◆ Hay diferentes formas de abordar el problema:

– datos comunes 
– mensajes.

Comunicación con datos comunes

◆ En un sistema monoprocesador, la forma más directa de 
comunicación entre dos o más procesos consiste en 
compartir datos
– un proceso escribe una variable compartida, el otro la lee

◆ Sin embargo, el acceso incontrolado a variables comunes 
puede producir resultados anómalos
– se dice que hay una condición de carrera cuando el resultado de la 

ejecución depende del orden en que se intercalan las instrucciones 
de dos o más procesos

– se trata de una situación anómala que hay que evitar

Ejemplo

T1 T2

contador=0

Incrementar 
contador

LDA contador
INC
STA contador

Incrementar 
contador

LDA contador
INC
STA contador

◆ El resultado final puede ser 1 ó 2
– Depende de las velocidades relativas de los procesos
– Para evitar condiciones de carrera hay que asegurar que las 

operaciones con variables comunes se ejecutan de forma atómica  
(indivisible)

Exclusión mutua

◆ Una secuencia de instrucciones que se debe ejecutar de 
forma indivisible se denomina sección crítica
– las operaciones de dos secciones críticas respecto a la misma 

variable no se deben entremezclar

◆ La forma de sincronización que se usa para proteger una 
sección crítica se llama  exclusión mutua
– supondremos que el acceso a una celda de memoria es atómico

(hay exclusión mutua entre operaciones que sólo hacen un acceso 
a memoria)

Sincronización condicional

◆ Cuando una acción de un proceso solo se puede ejecutar 
si otro proceso está en determinado estado o ha 
ejecutado ciertas acciones, hace falta un tipo de 
sincronización denominada sincronización condicional

◆ Ejemplo: Productor y consumidor con tampón limitado

– no se debe hacer Put cuando el tampón está lleno.
– no se debe hacer Get cuando el tampón está vacío
– además, hay exclusión mutua en el acceso al tampón

consumidortampónput getproductor

Espera ocupada

◆ Una forma de realizar la exclusión mutua es usar un 
indicador compartido
– Se puede hacer si el acceso al indicador sea atómico

(por ejemplo, mediante una instrucción test­and­set )

– Si no, hay que usar algoritmos complejos
◆ Es un método poco eficiente y complicado
◆ No es fácil imponer un orden de acceso cuando hay varios 

procesos esperando

while Test_and_Set(Flag) loop
   null;
end loop;
-- sección crítica
Flag := False;

Suspensión síncrona

◆ Una solución más eficiente consiste en disponer de un 
indicador que permita que un proceso se detenga o 
continúe su ejecución de forma atómica
– la atomicidad se asegura en el núcleo de multiprogramación

◆ Ejemplo

– ¡tiene una condición de carrera!
◆ Es mejor para sincronización condicional

if not Flag then
   suspend;
end if;
-- sección crítica
Flag := False;

Estados de un proceso

inexistente creado iniciándose ejecutable

terminado

esperando
inicio hijo

esperando
terminación pupilo suspendido

Semáforos

◆ Un semáforo es una variable que toma valores enteros no 
negativos

◆ Además de asignarle un valor inicial, sólo se pueden 
hacer dos operaciones con un semáforo S
– Down:  Si S > 0, decrementa S en 1 

Si S ≤ 0, la tarea se suspende hasta que S > 0
– Up:  Incrementa S en 1

Down y Up son indivisibles

◆ Otros nombres comunes para las operaciones de un semáforo son:
– Wait / Signal
– P / V  (passieren / vrijmachen) 
– Secure / Release,

Ejemplo

◆ Supongamos una definición como la siguiente
(ojo, no está predefinida en Ada)

◆ La implementación debe garantizar las propiedades 
de Down y Up

package Semaphores is
   type Semaphore (Initial : Integer := 1) 
     is limited private;
   
   procedure Down  (S : in out Semaphore);
   procedure Up    (S : in out Semaphore);

private
   ...
end Semaphores;

Exclusión mutua con semáforos

task T1;
task body T1 is
begin
  loop
    Wait(Mutex);
    -- sección crítica 1
    Up(Mutex);
    ...
   end loop;
 end T1;

task T2;
task body T2 is
begin
  loop
    Wait(Mutex);
    -- sección crítica 2
    Up(Mutex);
    ...
   end loop;
 end T2;

Mutex : Semaphore; -- iniciado a 1 por defecto

Mutex es un semáforo binario

Sincronización condicional

   task T1; -- espera
task body T1 is
begin
  loop
    S1A;
    Down(Condition);
    S1B;
   end loop;
 end T1;

   task T2; -- avisa
   task body T2 is
   begin
     loop
       S2A; -- establece condición
       Up(Condition);
       S2B;
     end loop;
   end T2;

Condition : Semaphore(0);

S1B no se ejecuta hasta que T2 avisa que se cumple la condición

Productor y consumidor (1)

package Buffer is  -- objeto abstracto
  procedure Put(X: in  Item);
  procedure Get(X: out Item);
end Buffer;

  package body Buffer is
    Size   : constant Positive := 32; -- por ejemplo
    subtype Index is mod Size;
    type    Store is array (Index) of Item; 
 
    Box         : Store;
    First, Last : Index := Index’First;
    Mutex       : Semaphore(1);    -- exclusión mutua
    Full;       : Semaphore(0);    -- elementos ocupados
    Empty       : Semaphore(Size); -- elementos libres

Productor y consumidor (2)

   procedure Put (X : in Item) is
begin
  Down(Empty);
  Down(Mutex);
  Box(Last) := X;
  Last        := Last + 1; 
  Up(Mutex);
  Up(Full);
end Put;

   procedure Get (X : out Item) is
begin
  Down(Full);
  Down(Mutex);
  X     := Box(First);
  First := First + 1;
  Up(Mutex);
  Up(Empty);
 end Get;

end Buffer;

Problemas de los semáforos

◆ Es fácil cometer errores
– Un solo Down o Up mal colocado puede hacer que el programa 

funcione incorrectamente
Ejemplos:

Up(S); ... Up(S);              ­­ no hay exclusión mutua
Down(S);   ... Down(S); ­­ interbloqueo
Up(S); ... Down(S); ­­ depende

◆ Es mejor usar mecanismos más abstractos y fiables
– regiones críticas
– monitores
– objetos protegidos

Interbloqueo (deadlock)

◆ Dos procesos están interbloqueados si cada uno de 
ellos espera un condición o un recurso que depende 
del otro

X,Y : Semaphore;

task T1; 
task body T1 is
begin
  loop
    Down(X);
    Down(Y);
    -- usar recursos
    Up(Y));
    Up(X);
 end loop;
 end T1;

task T2; 
task body T2 is
begin
  loop
    Down(Y);
    Down(X);
    -- usar recursos
    Up(X);
    Up(Y);
   end loop;
 end T2;

Interbloqueos, bloqueos “vivos” 
e inanición

◆ Se pueden evitar los interbloqueos asignando los recursos 
de forma ordenada
– no siempre es posible

◆ Se puede intentar asignar los recursos de forma más 
flexible
– pueden aparecer otros problemas:

» bloqueo “vivo”  (livelock)
» los procesos se ejecutan, pero ninguno consigue recursos
» inanición (starvation)
» los procesos se ejecutan, pero algunos no consiguen el recurso 

nunca

Regiones críticas condicionales

◆ Una región crítica es una secuencia de instrucciones que 
se ejecuta en exclusión mutua
– el compilador produce el código de bajo nivel necesario
– se identifica la variable compartida a la que se accede en una 

sección crítica
◆ La sincronización condicional se consigue mediante 

guardas  en las regiones críticas
– solo se puede entrar en la región crítica cuando la guarda es 

verdadera

Ejemplo (ojo, no es Ada):
V : shared T;
...

region V when condición is
   ...
end region;

Productor y consumidor (1)

package body Buffer is

  Size   : constant Positive := 32; -- por ejemplo
  subtype Index is mod Size;
  subtype Count is Natural range 0..Size;
  type    Store is array (Index) of Item; 
  
  type Buffer_Type is
    record
      Box         : Store;
      First, Last : Index := Index’First;
      Number      : Count := 0;
    end record;

  Data : shared Buffer_Type;  -- ¡no es Ada!  

Productor y consumidor (2)

   procedure Put (X : in Item) is
begin
  region Data when Data.Number < Size do  -- ¡no es Ada!
    Data.Box(Data.Last) := X;
    Data.Last            := Data.Last + 1;
    Data.Number          := Data.Number + 1;
  end region; 
end Put;

   procedure Get (X : out Item) is
begin
  region Data when Data.Number > 0 do     -- ¡no es Ada!
    X           := Data.Box(Data.First);
    Data.First  := Data.First + 1;
    Data.Number := Data.Number - 1;
  end region;
end Get;

end Buffer;

Problemas de las regiones críticas

◆ Implementación complicada
– cada vez que un proceso sale de una región crítica hay que volver 

a evaluar las guardas de los procesos que están esperando en 
otras regiones con la misma variable

– para ello hay que reanudar el proceso que espera, que se puede 
volver a suspender si la guarda es falsa

– potencialmente, muchos cambios de contexto

◆ No es obligatorio que estén confinadas en módulos
– el programa puede ser difícil de comprobar y mantener

◆ Una solución mejor son los monitores

Monitores

◆ Un monitor es un módulo cuyas operaciones se ejecutan 
en exclusión mutua
– el compilador produce el código de bajo nivel necesario

◆ La sincronización condicional se obtiene con
variables de condición
– dos operaciones primitivas: signal y wait

» wait suspende el proceso de forma incondicional y lo pone en una 
cola asociada a la condición que espera

» signal reanuda el primer proceso que espera en la señal
» no hay memoria: si no espera nadie, la señal se pierde

– un proceso que espera una condición libera el monitor

Productor y consumidor (1)

monitor Buffer is -- ¡ojo! no es Ada
  procedure Put(X: in  Item);
  procedure Get(X: out Item);
end Buffer;

  monitor body Buffer is -- ¡ojo! no es Ada

    Size   : constant Positive := 32; -- por ejemplo
    subtype Index is mod Size;
    subtype Count is Natural range 0..Size;
    type    Store is array (Index) of Item 

    Box         : Store;
    First, Last : Index := Index’First;
    Number      : Count := 0;
    Non_Full    : Condition := True;    -- ¡ojo! no es Ada 
    Non_Empty   : Condition := False;   -- ¡ojo! no es Ada 

Productor y consumidor (2)

   procedure Put (X : in Item) is
begin
  if Number = Size then
    Wait(Non_Full); -- ¡ojo! no es Ada 
  end if; 
  Box(Last) := X;
  Last        := Last + 1;
  Number      := Number + 1; 
  Signal(Non_Empty); -- ¡ojo! no es Ada 
end Put;

   procedure Get (X : out Item) is
begin
  if Number = 0 then
    Wait(Non_Empty); -- ¡ojo! no es Ada 
  end if; 
  X      := Box(First);
  First  := First + 1;
  Number := Number - 1;
  Signal(Non_Full); -- ¡ojo! no es Ada 
 end Get;

end Buffer;

Problemas de los monitores

◆ Las variables de condición son un mecanismo de bajo 
nivel

◆ La semántica de signal presenta problemas:
– Cuando un proceso hace signal puede haber dos procesos activos 

en el monitor
– Hay varias formas de evitar que se viole la exclusión mutua

» permitir signal solo si es la última instrucción de un subprograma
» forzar una salida del monitor al hacer signal
» suspender el proceso que hace signal si se reanuda otro proceso 

Objetos protegidos en Ada

◆ Combinación de monitores y regiones críticas 
condicionales

◆ Un objeto protegido es un objeto compuesto cuyas 
operaciones se ejecutan en exclusión mutua
– se pueden declarar tipos protegidos u objetos protegidos únicos
– en ambos casos hay especificación y cuerpo.

» la especificación contiene la interfaz visible desde otras unidades de 
programa.

– los tipos y objetos protegidos son unidades de programa, pero no 
de compilación.

» deben estar declarados en una unidad de compilación
(paquete o subprograma).

Ejemplo: entero compartido (1)

protected type Shared_Integer(Initial_Value : Integer) is

  function Value return Integer;
  procedure Change(New_Value : Integer);

private
  Data : Integer := Initial_Value;
end Shared_Integer;

◆ Las operaciones (subprogramas) se declaran en la parte pública
◆ Los detalles de implementación del tipo van en la parte privada

– solo son visibles en el cuerpo
– no se pueden declarar tipos de datos

◆ Es parecido a un registro
◆ Puede tener discriminantes

– tienen que ser de tipos discretos o acceso

Ejemplo: entero compartido (2)

protected body Shared_Integer is

  function Value return Integer is
  begin
    return Data;
  end Value;

  procedure Change(New_Value : Integer) is
  begin
    Data := New_Value;
  end Change;

end Shared_Integer;

◆ Los cuerpos de las operaciones van en el cuerpo del tipo protegido
◆ No se pueden declarar tipos ni objetos en el cuerpo

Ejemplo: entero compartido (3)

◆ No hay cláusula use para los objetos protegidos
◆ Las operaciones deben ir siempre cualificadas con el nombre del 

objeto

declare
  X : Shared_Integer(0);
  Y : Shared_Integer(1);
begin
  X.Change(Y.Value + 1); -- ahora X.Value es 2
end;

Diagramas de procesos

T1

P

op1

op2

op3

T2

◆ Se usan para representar la interacción entre tareas mediante 
objetos protegidos

Subprogramas protegidos

◆ Con los objetos protegidos solo se pueden efectuar 
operaciones protegidas
– Un procedimiento protegido da acceso exclusivo en lectura y 

escritura a los datos privados.
– Una función protegida da acceso concurrente en lectura sólo a los 

datos privados
◆ Se pueden ejecutar concurrentemente varias llamadas a 

una función protegida, pero no a un procedimiento 
protegido, ni a ambos a la vez.

◆ El núcleo de ejecución realiza la exclusión mutua 
mediante mecanismos más primitivos
– ¡ojo! una tarea que está intentando acceder a un objeto protegido 

no se considera suspendida
» la operaciones protegidas son cortas y no se pueden suspender

Entradas protegidas y 
sincronización condicional

◆ Una entrada es una operación protegida con una interfaz 
semejante a la de un un procedimiento

entry E (...);

◆ En el cuerpo se le asocia una barrera booleana

entry E (...) when B is ...

– si la barrera es falsa cuando se invoca la operación, la tarea que 
llama se suspende en una cola de espera

– cuando la barrera es verdadera, la tarea puede continuar
◆ Las entradas se usan para realizar sincronización 

condicional

Ejemplo: productor y consumidor (1)

  -- tampón limitado

  Size   : constant Positive := 10; -- por ejemplo
  type    Index is mod Size;
  subtype Count is Natural range 0 .. Size;
  type    Buffer_Store is array (Index) of Item;

  protected type Bounded_Buffer is
    entry Put(X : in  Item);
    entry Get(X : out Item);
  private
    First   : Index := Index’First;
    Last    : Index := Index'Last; 
    Number  : Count := 0; 
    Store   : Buffer_Store;
  end Bounded_Buffer;

  Buffer : Bounded_Buffer;

Productor y consumidor (2)

   protected body Bounded_Buffer is

  entry Put (X : in Item) when Number < Size is
  begin
    Last        := Last + 1;
    Store(Last) := X;
    Number      := Number + 1;
  end Put;

     entry Get (X : out Item) when Number > 0 is
  begin
    X           := Store(First);
    First       := First + 1;
    Number      := Number - 1;
  end Get;

end Bounded_Buffer;

Productor y consumidor (3)

  procedure Producer_Consumer is

  task Producer;
  task Consumer;

  task body Producer is
    X : Item;
  begin
    loop
      Produce(X);
      Buffer.Put(X);
    end loop;
  end Producer;

  task body Consumer is
    X : Item;
  begin
    loop
      Buffer.Get(X);
      Consume(X);
    end loop;
  end Consumer;

  end Producer_Consumer;

Evaluación de las barreras

◆ Las barreras se evalúan cuando
– una tarea llama a una entrada y la barrera hace referencia a una 

variable que puede haber cambiado desde la última evaluación
– una tarea termina la ejecución de un procedimiento o entrada y hay 

tareas esperando en entradas cuyas barreras hacen referencia a 
variables que pueden haber cambiado desde la última evaluación

◆ No se deben usar variables globales en las barreras

◆ La corrección de un programa no debe depender del 
momento en que se evalúan las barreras (se puede hacer 
con más frecuencia de lo indicado).

Ejemplo: control de recursos

  protected Resource_Control is
    entry Allocate;
    procedure Deallocate;
  private
    Free : Boolean := True;
  end Resource_Control;

  protected body Resource_Control is

    entry Allocate when Free is
    begin
      Free := False;
    end Allocate;

    procedure Deallocate is
    begin
      Free := True;
    end Deallocate;

  end Resource_Control;

Exclusión mutua y barreras

◆ Las tareas esperando en barreras tienen preferencia 
sobre las que esperan acceder al objeto
– de esta forma se evitan condiciones de carrera en la 

implementación
◆ Este modelo se llama cáscara de huevo (eggshell)

tareas esperando
entrar

tareas esperando
en barreras

una tarea dentro
como máximo

Restricciones

◆ En el cuerpo de una operación protegida no se pueden 
ejecutar operaciones "potencialmente bloqueantes":
– llamadas a entradas
– retardos
– creación o activación de tareas
– llamadas a subprogramas que contengan operaciones 

potencialmente bloqueantes
◆ Tampoco puede haber llamadas al mismo objeto

◆ El objetivo de estas restricciones es evitar que una tarea 
se quede suspendida indefinidamente en el acceso a un 
subprograma protegido

Ejemplo: radiado de mensajes (1)

  protected type Broadcast is
    entry     Receive (M : out Message);
    procedure Send    (M : in  Message);
  private
    New_Message     : Message;
    Message_Arrived : Boolean := False;
  end Broadcast;

Radiado de mensajes (2)

  protected body Broadcast is

    entry Receive (M : out Message) when Message_Arrived is
    begin
      M := New_Message;
      if Receive'Count = 0 then   -- ¡el último cierra!
        Message_Arrived := False;
      end if;
    end Receive;

    procedure Send (M : in  Message) is
    begin
      if Receive'Count > 0 then
        Message_Arrived := True;
        New_Message     := M;
      end if;
    end Send;

  end Broadcast;

Ejemplo: sucesos (1)

package Events is

  type Event is limited private;
  procedure Wait   (E : Event);
  procedure Signal (E : Event);

private

     protected type Event is
    entry Wait;
    procedure Signal;
  private

       Occurred : Boolean := False;
  end Event;

end Events;

Sucesos (2)

package body Events is

  protected body Event is

    entry Wait when Occurred is
    begin
      if Wait'Count = 0 then
        Occurred := False;
      end if;
    end Wait;
 
    procedure Signal is
    begin
      if Wait’Count > 0 then
        Occurred := True;
      end if;
    end Signal;

    end Event;

Sucesos (3)

   procedure Wait (E : Event) is
begin
  E.Wait;
end Wait;

procedure Signal (E : Event) is
begin
  E.Signal;
end Signal;

end Events;

Ejemplo: tarea esporádica

with Events; use  Events;
...
Button_Pressed : Event;
...
task body Sporadic is
begin
   loop
      Wait (Button_Pressed);
      -- acción esporádica
   end loop;
end Sporadic;

task body Event_Handler is
begin
   ...
   Signal (Button_Pressed); -- puede activar varias tareas
   ...
end Event_Handler;

Objetos de suspensión

◆ Proporcionan una funcionalidad similar cuando sólo hay 
una tarea que espera

package Ada.Synchronous_Task_Control is
   type Suspension_Object is limited private;
   procedure Set_True  (S : in out Suspension_Object);
   procedure Set_False (S : in out Suspension_Object);
   function  Current_State (S : Suspension_Object) 
     return Boolean;
   procedure Suspend_Until_True (S : in out Suspension_Object);
   -- Raises Program_Error if more than one task tries to
   -- suspend on S at once
   -- Sets S to False
private
   ...
end Ada.Synchronous_Task_Control;

Ejemplo: tarea esporádica

with Ada.Synchronous_Task_Control;
use  Ada.Synchronous_Task_Control;
...
Button_Pressed : Suspension_Object;

task body Sporadic is
   ...
begin
   loop
      Suspend_Until_True (Button_Pressed);
      -- acción esporádica
   end loop;
end Sporadic;

task body Event_Handler is
   ...
begin
   ...
   Set_To_True (Button_Pressed);
   ...
end Event_Handler;

Sincronización en C/POSIX.1c

◆ Hay dos mecanismos que permiten emular monitores con 
interfaz de procedimiento
– Un mutex es  una variable que proporciona exclusión mutua 

mediante dos operaciones, lock y unlock
– una variable de condición proporciona sincronización condicional 

mediante dos operaciones, wait y signal

◆ Ambos tipos de variables pueden tener atributos
◆ Otras operaciones: crear y destruir, espera temporizada

Definiciones

/* pthreads.h */

typedef ... pthread_mutext_t;
typedef ... pthread_cond_t;

...

int pthread_mutex_lock  (pthread_mutext_t *mutex);
int pthread_mutex_unlock(pthread_mutex_t *mutex);

int pthread_cond_wait(pthread_cond_t *cond ,
                      pthread_mutext_t *mutex);
   /* espera y libera el mutex */
int pthread_cond_signal(pthread_cond_t *cond);
   /* puede reanudar más de un thread */

...

Ejemplo: productor y consumidor (1)

#include <pthreads.h>
#define BUFF_SIZE 10

typedef struct {
  pthread_mutext_t mutex;
  pthread_cond_t buffer_non_full;
  pthread_cond_t buffer_non_empty;
  int count, first, last;
  int store[BUFF_SIZE];
} buffer_t;

Productor y consumidor (2)

int put(int item, buffer_t *buffer) {
  pthread_mutex_lock(&buffer->mutex);
  while (buffer->count == BUFF_SIZE)
    pthread_cond_wait(&buffer->non_full, &buffer->mutex);
  buffer[last] = item;
  last = (last++)%BUFF_SIZE; count++;
  pthread_cond_signal(&buffer->non_empty);
  pthread_mutex_unlock(&buffer->mutex);
  return 0;
}

int get(int *item, buffer_t *buffer) {
  pthread_mutex_lock(&buffer->mutex);
  while (buffer->number == 0)
    pthread_cond_wait(&buffer->non_empty, &buffer->mutex);
  item = buffer[first];
  first = (first++)%BUFF_SIZE; count--;
  pthread_cond_signal(&buffer->non_full);
  pthread_mutex_unlock(&buffer->mutex);
  return 0;
}

/* ¡ojo! hay que iniciar el mutex y las variables de condición */

Resumen

◆ El acceso de dos o más tareas concurrentes a las 
variables comunes debe realizarse en exclusión mutua

◆ Otra forma de sincronización está ligada al cumplimento 
de una condición

◆ Algunos mecanismos que permiten conseguir ambas 
formas de sincronización son:
– semáforos
– regiones críticas condicionales
– monitores

◆ Los objetos protegidos reúnen las ventajas de las 
regiones críticas condicionales y los monitores

◆ Los mutex y las variables condicionales de POSIX 
proporcionan un mecanismo similar a los monitores


